********************************************************************************
*
**#	Table 7: Does DV Court Reduce Re-Victimization?	
*		
********************************************************************************

use "${DATA}/data_VICTIMS.dta", clear

* 		Define Controls for Victims Level Analysis: 

gl v_cntrls			///
		victim_age prev_v prev_w i.judge_male c.judge_exp
		
* 		FEs and Cluster for Victims Level Analysis: 

gl v_hdfe_opts 		///	
		absorb(week_number year_court) vce(cluster def_id week_year) 
		
		
# d ; 

eststo clear ;
local m = 1  ;

	
foreach y of varlist 
				revict_3 			
				revict_3_aslt
				revict_3_warrant 
				refused_coop { ;
 
	eststo v_`m': 
	reghdfe `y' dv_judge , $v_hdfe_opts  ;
	
		estadd ysumm			;
		estadd local cntrl ""	;
		local m `++m'			;

	eststo v_`m': 
	reghdfe `y' dv_judge $v_cntrls , $v_hdfe_opts ;
	
		estadd ysumm			;
		estadd local cntrl "X"	;
		local m `++m'			;
} ;

	eststo v_`m': 
	reghdfe refused_coop_cond dv_judge if revict_3 == 1 , $v_hdfe_opts ;
	
		estadd ysumm			;
		estadd local cntrl ""	;
		local m `++m'			;

	eststo v_`m': 
	reghdfe refused_coop_cond dv_judge $v_cntrls if revict_3 == 1,$v_hdfe_opts ;
	
		estadd ysumm			;
		estadd local cntrl "X"	;
		local m `++m'			;

esttab v_* using "${OUTPUT}/tab7.rtf", replace ${tab_optns} 
		keep(dv_judge)
		scalars(
			"cntrl Controls"
			"ymean Y Mean")
		title( {\b Table 7. Does DV Court Reduce Re-Victimization?});

# d cr

********************************************************************************
*   End of File
********************************************************************************
